
Abstract

Spivey has recently presented a novel functional representation
that supports the efficient composition, or merging, of coroutine
pipelines for processing streams of data. This representation was
inspired by Shivers and Might’s three-continuation approach and is
shown to be equivalent to a simple yet inefficient executable spec-
ification. Unfortunately, neither Shivers and Might’s original work
nor the equivalence proof sheds much light on the underlying prin-
ciples allowing the derivation of this efficient representation from its
specification.

This paper gives the missing insight by reconstructing a sys-
tematic derivation in terms of known transformation steps from the
simple specification to the efficient representation. This derivation
sheds light on the limitations of the representation and on its appli-
cability to other settings. In particular, it has enabled us to obtain
a similar representation for pipes featuring two-way communication,
similar to the Haskell pipes library. Our benchmarks confirm that
this two-way representation retains the same improved performance
characteristics.

1 Introduction

Coroutine pipelines provide a compositional approach to processing streams of
data that is both efficient in time and space, thanks to a targeted form of lazy
evaluation interacting well with side-effects like I/O. Two prominent Haskell
libraries for coroutine pipelines are pipes [5] and conduit [12]. Common to both
libraries is their representation of pipelines by an algebraic data type (ADT).

Spivey [13] has recently presented a novel Haskell representation that is en-
tirely function-based. His representation is an adaptation of Shivers and Might’s
earlier three-continuation representation [11] and exhibits a very efficient merge
operation for connecting pipes.

Spivey proves that his representation is equivalent to a simple ADT-based
specification. Yet, neither his proof nor Shivers and Might’s explanation sheds
much light on the underlying principles used to come up with the efficient rep-
resentation. This makes it difficult to adapt the representation to other settings.

This paper remedies the situation by systematically deriving the efficient
function-based representation from the simple, but inefficient ADT-based repre-
sentation. Our derivation consists of known transformations and constructions

that are centered around folds with appropriate algebras. Our derivation clarifies
the limitations of the efficient representation, and enables us to derive a similarly
efficient represenation for the two-way pipes featured in the pipes library.

The specific contributions of this paper are:

– We present a systematic derivation of Spivey’s efficient representation start-
ing from a simple executable specification. Our derivation only consists of
known transformations, most of which concern structural recursion with folds
and algebras. It also explains why the efficient representation only supports
the merging of “never-returning” pipes.

– We apply our derivation to a more general definition of pipes used by the
pipes library, where the communication between adjacent pipes is bidirec-
tional rather than unidirectional.

– Our benchmarks demonstrate that the merge operator for the bidirectional
three-continuation approach improves upon the pipes library’s performance.

The rest of this paper is organized as follows. Section 2 briefly introduces
both the ADT pipes encoding and the three-continuation approach. Section 3
derives the fast merging operation for a simplified setting. Section 4 derives the
fast merging operation for the original pipe setting. Section 5 extends Spivey’s
approach with the bidirectional pipes operations. Section 6 presents the results
of the primes benchmark by Spivey, on the approaches discussed in this paper.
Section 7 discusses related work and Section 8 concludes this paper.

2 Motivation

This section introduces the ADT pipes encoding and then contrasts it with the
three-continuation encoding. This serves as both a background introduction and
a motivation for a better understanding of the relation between both encodings.

2.1 Pipes

We start with a unidirectional version of the pipes library. A unidirectional pipe
can receive i values, output o values and return a values. On the other hand, a
bidirectional pipe additionally carries an output value when receiving values and
an input value when outputting values. We represent a unidirectional pipe as an
abstract syntax tree where each node is an input, output or return operation.
This is expressed in Haskell with the following ADT.

data Pipe i o a = Input (i → Pipe i o a)
| Output o (Pipe i o a)
| Return a

This datatype exhibits a monadic structure where the bind operation (>>=) grafts
one syntax tree onto another.

instance Monad (Pipe i o) where
return = Return
(Input h) >>= f = Input (λi → (h i)>>= f )
(Output o r)>>= f = Output o (r >>= f )
(Return a) >>= f = f a

We define the basic components: inputP : a pipe returning its received input,
outputP : a pipe outputting a set value and returnP : a pipe returning a set value.

inputP :: Pipe i o i
inputP = Input (λi → Return i)

outputP :: o → Pipe i o ()
outputP o = Output o (Return ())

returnP :: a → Pipe i o a
returnP a = Return a

The bind operation assembles these componenents into larger pipes. For exam-
ple doublerP , a pipe which repeatedly takes its input, multiplies it by two and
continually outputs this new value.

doublerP :: Pipe Int Int a
doublerP = do i ← inputP ; outputP (i ∗ 2 ); doublerP

Another essential way of combining pipes is merging them. This connects the
outputs of the upstream to the inputs of the downstream. In the implementation,
mergePL performs a case analysis on the downstream q : if it is trying to output
then that is kept and we keep searching, if it finds an input then we call mergePR

on the wrapped continuation and the upstream. Then, in mergePR we similarly
scan the upstream for an output operation, keeping any input operations. If an
output operation is found, the output value is passed to the continuation and
the merging process starts again. If at any point we see a return, then the merge
finishes with this resulting return value. The implementation is given below.

mergeP :: Pipe i m a → Pipe m o a → Pipe i o a
mergeP p q = mergePL q p where

mergePL :: Pipe m o a → Pipe i m a → Pipe i o a
mergePL (Input h) p = mergePR p h
mergePL (Output o r) p = Output o (mergePL r p)
mergePL (Return a) p = Return a

mergePR :: Pipe i m a → (m → Pipe m o a)→ Pipe i o a
mergePR (Input f ) h = Input (λv → mergePR (f v) h)
mergePR (Output o r) h = mergePL (h o) r
mergePR (Return a) h = Return a

The merge operator enables expressing the merge of doublerP with itself. In
this example the left doublerP is the upstream and the right doublerP is the

downstream. The result of this merge is a pipe which outputs the quadruple of
its incoming values.

quadruplerP :: Pipe Int Int a
quadruplerP = doublerP ‘mergeP ‘ doublerP

We can run a pipe by interpreting it to IO .

toIOP :: (Read i ,Show o)⇒ Pipe i o a → IO a
toIOP (Input f ) = do i ← readLn; toIOP (f i)
toIOP (Output o r) = do putStrLn ("out: " ++ show o); toIOP r
toIOP (Return a) = return a

An example where we input 10 , receive 40 and then exit, is shown below.

λ > toIOP quadruplerP
10 〈Return〉
out : 40
〈Ctrl+C〉

2.2 Three-Continuation Approach

The function mergeP is suboptimal because it has to recursively scan a pipe for
an operation of interest while copying the other operation. When several merges
are piled up this leads to repeated scanning and copying of the same operations.

Spivey has introduced ContPipe, a different pipe representation which en-
ables a faster merge implementation [13]. It featuresthree continuations, one for
each constructor. The first continuation (a → Result i o) represents the return
constructor. The next two continuations, InCont i and OutCont o as part of
Result i o, represent the input and output constructor respectively.

newtype ContPipe i o a =
MakePipe {runPipe :: (a → Result i o)→ Result i o}

type Result i o = InCont i → OutCont o → IO ()

newtype InCont i =
MakeInCont {resumeI :: OutCont i → IO ()}

newtype OutCont o =
MakeOutCont {resumeO :: o → InCont o → IO ()}

instance Monad (ContPipe i o) where
return a = MakePipe (λk → k a)
p >>= f = MakePipe (λk → runPipe p (λx → runPipe (f x ) k))

In the following definitions for the basic pipe components the continuation k is
the return constructor—we give it a value and the input and output constructors
and receive a pipe. The continuations ki and ko are the input and output con-
structors, we resume them with the newtype unwrapper and the continuations
are refreshed once they have been used.

returnCP :: a → ContPipe i o a
returnCP a = MakePipe (λk ki ko → k a ki ko)

inputCP :: ContPipe i o i
inputCP = MakePipe (λk ki ko →

resumeI ki (MakeOutCont (λi k ′i → k i k ′i ko)))

outputCP :: o → ContPipe i o ()
outputCP o = MakePipe (λk ki ko →

resumeO ko o (MakeInCont (λk ′o → k () ki k ′o)))

We can use the Monad instance for ContPipe to compose pipes with do-notation,
similar to Pipe.

doublerCP :: ContPipe Int Int a
doublerCP = do i ← inputCP ; outputCP (i ∗ 2 ); doublerCP

We can also interpret ContPipe to IO .

toIOCP :: (Read i ,Show o)⇒ ContPipe i o ()→ IO ()
toIOCP p = runPipe p (λ() → return ()) ki ko where

ki = MakeInCont (λko → do x ← readLn; resumeO ko x ki)
ko = MakeOutCont (λo ki →

do putStrLn ("out: " ++ show o); resumeI ki ko)

The merge function for ContPipe is defined as:

mergeCP p q = MakePipe (λk ki ko →
runPipe q err (MakeInCont (λk ′o → runPipe p err ki k ′o)) ko)
where err = error "terminated"

With the merge definition we are able to create the quadrupler pipe as before.
Running toIOCP quadruplerCP results an identical scenario to the Pipe scenario
from the previous section.

quadruplerCP :: ContPipe Int Int a
quadruplerCP = doublerCP ‘mergeCP ‘ doublerCP

While Spivey has demonstrated the remarkable performance advantage of
this merge operator, he sheds little light on the origin or underlying principles of
the related encoding. The remainder of this paper provides this missing insight
by deriving Spivey’s efficient ContPipe representation from the ADT-style Pipe
by means of well-known principles. The aim is to improve understanding of the
applicability and limitations of the techniques used.

3 Fast Merging for One-Sided Pipes

To offer a firmer grip on the problem, this section considers a simplified setting
where pipes are one-sided, either only producing or only consuming data. For
example, the doubler component can not be defined in this setting. The simplified
setting gives a more straightforward path to the fast merging approach, which
we generalize back to regular ‘mixed’ pipes in Section 4.

3.1 One-Sided Pipes

In the simplified setting pipes are either pure Producers or pure Consumers. A
Producer only outputs values, while a Consumer only receives them.

data Producer o = Producer o (Producer o)
data Consumer i = Consumer (i → Consumer i)

If we specialize mergeP for a Consumer and a Producer , we get:

mergeA :: Producer b → Consumer b → a
mergeA p q = mergeAL q p where

mergeAL :: Consumer b → Producer b → a
mergeAL (Consumer h) p = mergeAR p h

mergeAR :: Producer b → (b → Consumer b)→ a
mergeAR (Producer o r) h = mergeAL (h o) r

3.2 Mutual Recursion Elimination

The two auxiliary functions mergeAL and mergeAR turn respectively a producer
and a consumer into the result of type a by means of an additional parameter,
which is respectively of type (Producer b) and (b → Consumer b). To highlight
these parameters, we introduce type synonyms for them.

type ProdPar ′ b = Producer b
type ConsPar ′ b = b → Consumer b

Now we refactor mergeAL and mergeAR with respect to their additional parame-
ter in a way that removes the term-level mutual recusion between them. Consider
mergeAL which does not use its parameter p directly, but only its interpretation
by function mergeAR. We refactor this code to a form where mergeAR has al-
ready been applied to p before it is passed to mergeAL. This adapted mergeAL

would then have type Consumer b → (ConsPar ′ b → a) → a. At the same
time we apply a similar transformation to mergeAR, moving the application of
mergeAL to h out of it. This yields infinite types for the two new parameters,
which Haskell only accepts if we wrap them in newtypes.

newtype ProdPar b a = ProdPar (ConsPar b a → a)
newtype ConsPar b a = ConsPar (b → ProdPar b a → a)

The merge is then defined by appropriately placing newtype (un-)wrappers.

mergePar :: Producer b → Consumer b → a
mergePar p q = ml q (ProdPar (mr p)) where

ml :: Consumer b → ProdPar b a → a
ml (Consumer h) (ProdPar p) = p (ConsPar (λi → (ml (h i))))

mr :: Producer b → ConsPar b a → a
mr (Producer o r) (ConsPar h) = h o (ProdPar (mr r))

Note that we can recover Spivey’s InCont i and OutCont o by instantiating the
type parameter a to IO () in ProdPar i a and ConsPar o a respectively.

3.3 Structural Recursion with Fold

Due to the removal of the term-level mutual recursion in ml and mr , they are
easily adapted to their structurally recursive form. By isolating the work done
in each recursive step, we obtain algL and algR.

type CarrierL i a = ProdPar i a → a

algL :: (i → CarrierL i a)→ CarrierL i a
algL f = λ(ProdPar prod)→ prod (ConsPar f )

type CarrierR o a = ConsPar o a → a

algR :: o → CarrierR o a → CarrierR o a
algR o prod = λ(ConsPar cons)→ cons o (ProdPar prod)

The functions algL and algR are now in a form known as algebras. Algebras are a
combination of a carrier r , the type of the resulting value, and an action of type
f r → r . This action denotes the computation performed at each node of the
recursive datatype, for which the functor f determines the shape of its nodes.
We omit the carrier type if it is clear from the context and simply refer to an
algebra by its action.

The structural recursion schemes, or folds, for Consumer and Producer take
algebras of the form (i → r)→ r and o → r → r . Their definitions are:

foldP :: (o → r → r)→ Producer o → r
foldP alg (Producer o r) = alg o (foldP alg r)

foldC :: ((i → r)→ r)→ Consumer i → r
foldC alg (Consumer h) = alg (λi → foldC alg (h i))

An example use of folds is an interpretation to IO by supplying the inputs for a
consumer or printing the outputs of a producer.

type CarrierConsIO i = IO ()

consumeIO :: Read i ⇒ Consumer i → IO ()
consumeIO = foldC alg where

alg :: Read i ⇒ (i → CarrierConsIO i)→ CarrierConsIO i
alg f = do x ← readLn; f x

type CarrierProdIO o = IO ()

produceIO :: Show o ⇒ Producer o → IO ()
produceIO = foldP alg where

alg :: Show o ⇒ o → CarrierProdIO o → CarrierProdIO o
alg o p = do print o; p

Another example is expressing mergePar with folds using algL and algR.

mergefold :: Producer x → Consumer x → a

mergefold p q = foldC algL q (ProdPar (foldP algR p))



8 R. P. Pieters and T. Schrijvers

3.4 A Short-Cut to a Merge-Friendly Representation

Instead of directly defining a Consumer or Producer value in terms of the data
constructors of the respective types, we can also do it in a more roundabout way
by abstracting over the constructor occurrences—this is known as build form.
The build function then instantiates the abstracted constructors with the actual
constructors; for Consumer and Producer they are:

buildC :: (∀r .((i → r)→ r)→ r)→ Consumer i
buildC g = g Consumer

buildP :: (∀r .(o → r → r)→ r)→ Producer o
buildP g = g Producer

For instance,

prodFrom :: Integer → Producer Integer
prodFrom n = Producer n (prodFrom (n + 1 ))

can be written as:

prodFrom n = buildP (prodFrom ′ n) where
prodFrom ′ :: Integer → (∀r .(Integer → r → r)→ r)
prodFrom ′ n p = go n where go n = p n (go (n + 1 ))

The motivation for these build functions is use of the fold/build fusion rule, a
special form of short-cut fusion [4]. This rule can be applied when a fold directly
follows a build, specifically for Consumer and Producer these fusion rules are:

foldC alg (buildC cons) = cons alg

foldP alg (buildP prod) = prod alg

In other words, instead of first building an ADT representation and then folding
it to its result, we can directly create the result of the fold. This readily applies to
the two folds in mergefold. We can directly represent consumers and producers
in terms of the carrier types of those two folds,

type ConsumerAlt i = ∀a.CarrierL i a -- ∀a.ProdPar i a → a
type ProducerAlt o = ∀a.CarrierR o a -- ∀a.ConsPar o a → a

using their algebras as constructors:

inputAlt :: (i → ConsumerAlt i)→ ConsumerAlt i
inputAlt = algL
outputAlt :: o → ProducerAlt o → ProducerAlt o
outputAlt = algR

For instance, after fold/build fusion prodFrom becomes:

prodFromAlt :: Integer → ProducerAlt Integer
prodFromAlt n = outputAlt n (prodFromAlt (n + 1 ))

The merge function for the alternate representations ProducerAlt and ConsumerAlt

then becomes an almost trivial operation.

mergeAlt :: ProducerAlt b → ConsumerAlt b → a
mergeAlt p q = q (ProdPar p)

3.5 A Not So Special Representation

This merge-friendly representations of producers and consumers are not just
specializations; they are in fact isomorphic to the originals. The inverses of ml
and mr to complete the isomorphism are given by ml−1 and mr−1. Appendix A
proves the isomorphism.

ml−1 :: ConsumerAlt i → Consumer i

ml−1 f = f (ProdPar h) where
h :: ConsPar i (Consumer i)→ Consumer i
h (ConsPar f ) = Consumer (λx → f x (ProdPar h))

mr−1 :: ProducerAlt o → Producer o
mr−1 f = f (ConsPar (λx p → Producer x (h p))) where

h :: ProdPar o (Producer o)→ Producer o
h (ProdPar f ) = f (ConsPar (λx p → Producer x (h p)))

Hence, we can also fold with other algebras by transforming the merge-friendly
representation back to the ADT, and then folding over that.

foldPAlt
:: (o → a → a)→ ProducerAlt o → a

foldPAlt
alg rep = foldP alg (mr−1 rep)

foldCAlt
:: ((i → a)→ a)→ ConsumerAlt i → a

foldCAlt
alg rep = foldC alg (ml−1 rep)

Of course, these definitions are wasteful because they create intermediate datatypes.
However, by performing fold/build fusion we obtain their fused versions:

foldPAlt
alg rep = rep (ConsPar (λx p → alg x (h p))) where

h (ProdPar f ) = f (ConsPar (λx p → alg x (h p)))

foldCAlt
alg rep = rep (ProdPar h) where

h (ConsPar f ) = alg (λx → f x (ProdPar h))

4 Return to Two-Sided Pipes

The previous section has derived an efficient approach for simplified Consumer
and Producer pipes. This section extends that approach to proper Pipes in two
steps, first supporting both input and output operations, and then also a return.

4.1 Pipe of No Return

Let us consider pipes with both input and output operations, but no return.

data Pipe∞ i o = Input∞ (i → Pipe∞ i o)
| Output∞ o (Pipe∞ i o)

We can fold over these pipes by providing algebras for both the input and output
operation, agreeing on the carrier type a.

foldPipe∞ :: Pipe∞ i o → ((i → a)→ a)→ (o → a → a)→ a
foldPipe∞ p inAlg outAlg = go p where

go (Input∞ p) = inAlg (λi → go (p i))
go (Output∞ o p) = outAlg o (go p)

To merge these pipes, we use algL and algR developed in the previous section.
There is only one snag: the two algebras do not agree on the carrier type. The
carrier types were the alternate representations ConsumerAlt and ProducerAlt .

type ConsumerAlt i = ∀a.ProdPar i a → a
type ProducerAlt o = ∀a.ConsPar o a → a

We reconcile these two carrier types by observing that both are functions with
a common result type, but different parameter types. A combination of both is
a function taking both parameter types as input.

type ResultR i o = ∀a.ConsPar o a → ProdPar i a → a

The algebra actions are easily adapted to the additional parameter. They simply
pass it on to the recursive positions without using it themselves.

inputResultR :: (i → ResultR i o)→ ResultR i o
inputResultR f = λcons (ProdPar prod)→

prod (ConsPar (λi prod ′ → f i cons prod ′))

outputResultR :: o → ResultR i o → ResultR i o
outputResultR o result = λ(ConsPar cons) prod →

cons o (ProdPar (λcons ′ → result cons ′ prod))

Like before, we can avoid the algebraic datatype Pipe∞ and directly work with
ResultR using the algebras as constructor functions.

Finally, we can use the one-sided merge function from the previous section to
merge the output side of a ResultR i m pipe with the input side of a ResultR m o
pipe. Because we defer the interpretation of the i and o sides of the respective
pipes, this one-sided merge does not yield a result of type a, but rather one of
type ConsPar o a → ProdPar i a → a. In other words, the merge of the two
pipes yields a ResultR i o pipe.

mergeResultR :: ResultR i m → ResultR m o → ResultR i o
mergeResultR p q = λconso prodi →

let q ′ = q conso
p′ = flip p prodi

in q ′ (ProdPar p′)

4.2 Return to return

Finally, we reobtain return and the monadic structure of pipes in a slightly
unusual way, by means of the continuation monad.

newtype Cont r a = Cont {runCont :: (a → r)→ r }
instance Monad (Cont r) where

return x = Cont (λk → k x )
p >>= f = Cont (λk → runCont p (λx → runCont (f x ) k))

If we specialize the result type r to ResultR i o, we get:

newtype ContP i o a = ContP ((a → ResultR i o)→ ResultR i o)

The merge function for ContP is implemented in terms of mergeResultR .

mergeCont :: ContP i m Void → ContP m o Void → ContP i o a
mergeCont (ContP p) (ContP q) = ContP (λk →

mergeResultR (p absurd) (q absurd))

However, there is an issue: before mergeResultR can merge the two pipes, their
continuations (the interpretations of the return constructor) must be supplied.
Yet, the resulting pipe’s continuation type k does not match that of either the
upstream or downstream pipe. Thus we are stuck, unless we assume what we
have been all along: that the two pipes are infinite. Indeed, in that case it does
not matter that we don’t have a continuation for them, as their continuation is
never reached anyway. In short, mergeCont only works for never-returning pipes,
which we signal with the return type Void , only inhabited by ⊥.

4.3 Specialization for IO

To get exactly Spivey’s representation, we instantiate the polymorphic type vari-
able a in ResultR i o to IO (), which yields:

type Result i o = InCont i → OutCont o → IO ()

We can rewrite this type as a monad transformer stack, using two reader monad
transformers for the two parameters.

newtype ReaderT r m a = ReaderT {runReaderT :: r → m a }
type Result ′ i o = ReaderT (InCont i) (ReaderT (OutCont o) IO) ()

Similarly, ContPipe i o a can be written with a transformer stack by adding a
ContT layer, since Cont (m r) is equal to ContT r m for any monad m.

newtype ContT r m a = ContT {runContT :: (a → m r)→ m r }
type ContPipe ′ i o a =

ContT () (ReaderT (InCont i) (ReaderT (OutCont o) IO)) a

This transformer stack view enables two additional useful operations: aborting
the pipe and embedding an IO action. Both are specializations of generic func-
tionality from the continuation monad transformer: abort and liftContT .

abort :: m r → ContT r m a
abort r = ContT (λk → r)

liftContT :: Monad m ⇒ m a → ContT r m a
liftContT p = ContT (λk → p >>= k)

exit ′ :: ContPipe ′ i o a
exit ′ = abort (liftReaderT (liftReaderT (return ())))

effect ′ :: IO a → ContPipe ′ i o a
effect ′ e = liftContT (liftReaderT (liftReaderT e))

5 Bidirectional Pipes

So far we have covered unidirectional pipes where information flows in one di-
rection through the pipe, from the output operations in one pipe to the input
operations in the next pipe downstream. However, some use cases also require
information to flow upstream and pipes that support this are called bidirectional.

The Proxy data type at the core of the pipes library [5] implements bidirec-
tional pipes. The operations request and respond are respectively downstream
and upstream combinations of input and output . In addition, Proxy is also a
monad transformer that embed effects of monad m.

data Proxy a ′ a b′ b m r = Request a ′ (a → Proxy a ′ a b′ b m r)
| Respond b (b′ → Proxy a ′ a b′ b m r)
| M (m (Proxy a ′ a b′ b m r))
| Pure r

We refer to the pipes source code [5] for the implementation of the corresponding
mergePL and mergePR functions, which are called +>> and >>∼.

We obtain a more efficient function-based representation by adapting the
derivation of Sections 3 and 4. This yields the parameter type PCPar .

newtype PCPar i o a = PCPar (o → PCPar o i a → a)

The ResultR counterpart for Proxy takes two such PCPars as input. In addition,
the result type r is now a monadic type m r to be able to lift operations once it
is wrapped with Cont .

type ProxyRep a ′ a b′ b m = ∀r .PCPar a a ′ (m r)→ -- request
PCPar b′ b (m r)→ -- respond
m r

Then, we can proceed with defining the merge function for ProxyRep and the
Cont-wrapped version similar to ResultR.

mergeProxyRep :: (c′ → ProxyRep a ′ a c′ c m)→ ProxyRep c′ c b′ b m →
ProxyRep a ′ a b′ b m

mergeProxyRep fc′ q = λreq res →
let p′ c′ = fc′ c′ req

q ′ = flip q res
in q ′ (PCPar p′)

newtype ContPr a ′ a b′ b m r = ContPr {unContPr ::
(r → ProxyRep a ′ a b′ b m)→ ProxyRep a ′ a b′ b m }

mergeContPr :: (c′ → ContPr a ′ a c′ c m Void)→
ContPr c′ c b′ b m Void → ContPr a ′ a b′ b m r

mergeContPr fc′ (ContPr q) = ContPr (λk →
mergeProxyRep (λc′ → unContPr (fc′ c′) absurd) (q absurd))

6 Benchmarks

Figure 1 shows the results of Spivey’s primes benchmark, which calculates the
first n primes. The benchmarks are executed using the criterion library [9] on
an Intel Core i7-6600U at 2.60 GHz with 8 GB memory running Ubuntu 16.04
and GHC 8.4.3, with -O2 enabled.1

The figure compares the pipes (v4.3.9) and conduit (v1.3.0.3) libraries to
Spivey’s original implementation (contpipe) and our generalized form (proxyrep).

We can see that the former two libraries, which use an ADT representa-
tion, both show the quadratic performance behaviour for a use case with a high
amount of merge steps. On the other hand, the latter two show the improved
performance behaviour. The slight overhead of proxyrep compared to contpipe

can be explained by the specialization to IO () in the latter type.
Appendix B contains the results of some additional microbenchmarks.

7 Related Work

We have covered the main related works of Spivey [13], Shivers and Might [11]
and the pipes library [5] in the body of the paper. Below we discuss some
additional related work.

Encodings The Church [2,1] and Scott [8] encodings encode ADTs using func-
tions. The encoding derived in this paper has a close connection to the Scott en-
coding. The Scott encoding for Producer and Consumer are ScottP and ScottC .
By moving the quantified variable a to the definition, we obtain SP and SC .

newtype ScottP o = ScottP (∀a.(o → ScottP o → a)→ a)
newtype ScottC i = ScottC (∀a.((i → ScottC i)→ a)→ a)

1 The benchmarks are at https://github.com/rubenpieters/orth-pipes-bench.

https://github.com/rubenpieters/orth-pipes-bench

Fig. 1. Results of the primes benchmark.

newtype SP o a = SP ((o → SP o a → a)→ a)
newtype SC i a = SC (((i → SC i a)→ a)→ a)

Then, ∀a.SP o a is representationally equivalent to ProducerAlt and similarly
for ∀a.SC i a and ConsumerAlt (see Appendix C).

If we look at the Scott encoding ScottPipe∞ for Pipe∞, we can obtain an
equivalent representation to ResultR by using SP and SC instead of ScottPipe∞
in the parameter corresponding to their operations.

newtype ScottPipe∞ i o = ScottPipe∞
(∀a.(o → ScottPipe∞ i o → a)→ ((i → ScottPipe∞ i o)→ a)→ a)

type SP∞ i o = ∀a.(o → SP o a → a)→ ((i → SC i a)→ a)→ a

We dubbed this the orthogonal encoding due to the separation of the operations.

Conduit The conduit library [12] is another popular choice for Haskell stream
processing. The two main differing points of conduit with pipes is a built-
in representation of leftovers and detection of upstream finalization. Leftovers
are operations representing unprocessed outputs. For example in a takeWhile
pipe, which takes outputs untill a condition is matched, the first element not
matching the condition will also be consumed. This element can then be emitted
as a leftover, which will be consumed by the downstream with priority. Detecting
upstream finalization is handled by input returning Maybe values, where Nothing
represents the finalization of the upstream.

Parsers Spivey mentioned in his work [13] that the ContPipe approach might
be adapted to fit the use case of parallel parsers [3]. However, after gaining more
insight into ContPipe, it does not seem that the merging operation for parsers
immediately fits the pattern presented in this paper. One of the problematic
elements is the fail operation, which is not passed as-is to the newly merged
structure, but given a non-trivial interpretation. Namely, an interpretation de-
pendent on the other structure during the recursive merge process.

Shallow To Deep Handlers The handlers framework by Kammar et al. [6] sup-
ports both shallow handlers, based on case analysis, and deep handlers, based on
folds. They cover an example of transforming a producer and consumer merging
function from shallow handlers to deep handlers. This example is related to our
simplified setting in Section 3. To do this they introduce Prod and Cons, which
are equivalent to our ProdPar and ConsPar . Compared to their example, we
take a more step-by-step explanatory approach and additionally move to more
complicated settings in our further sections.

Multihandlers The Frank language [7] is based on shallow handlers and sup-
ports a feature called multihandlers. These handlers operate on multiple inputs
which have uninterpreted operations, much like pattern matching on multiple
free structures. The patterns we have handled in this paper are concerned with
pattern matching on multiple data structures and a mutual relation between
these functions. This seems like an interesting connection to investigate further.

8 Conclusion

We have given an in-depth explanation of the principles behind the fast merging
of the three-continuation approach. We have given a series of steps to derive this
fast implementation from the less efficient one.

We apply this pattern to the setting of bidirectional pipes, as in the pipes

library. This results in a more general version of this representation, but still has
the same performance due to its efficient merge implementation.

We apply Spivey’s benchmarks [13] to check that our generalized encoding
retains similar performance. We also include the pipes library in the benchmark
to compare with a commonly used implementation of bidirectional pipes.

This pipes encoding has been made available as a library on github2.

2 https://github.com/rubenpieters/Orthogonal-Pipes

https://github.com/rubenpieters/Orthogonal-Pipes

A Proof isomorphism Producer and ProducerAlt

We show the Producer case, an analogous strategy works for Consumer .

Theorem 1. The types ProducerAlt o and Producer o are isomorphic.

Proof. We show this by using µx. G [x ] = ∀x .(G [x ]→ x )→ x [10].

ProducerAlt o
= (see Appendix C)
∀r. SP o r

= (def. SP)
∀r.µx. (o → x → r)→ r

= (property µ)
∀r.∀x. (((o → x → r)→ r)→ x )→ x

= (swap ∀)
∀x.∀r. (((o → x → r)→ r)→ x )→ x

= (property µ)
µx.∀r. (o → x → r)→ r

= (uncurry)
µx.∀r. ((o, x )→ r)→ r

= (property µ)
µx.µr. (o, x )

= (drop unused µ)
µx. (o, x )

= (def. Producer)
Producer o ut

An isomorphism consists of two inverses, here mr and mr−1, for which mr−1◦
mr = id and mr ◦mr−1 = id holds. Where mr is defined as:

mr :: Producer x → ProducerAlt x
mr (Producer o r) (ConsPar h) = h o (ProdPar (mr r))

And mr−1 is defined as:

mr−1 :: ProducerAlt x → Producer x

mr−1 f = f (ConsPar (λx p → Producer x (helpermr−1 p)))

helpermr−1 :: ProdPar x (Producer x )→ Producer x

helpermr−1 (ProdPar f ) = f (ConsPar (λx prod → Producer x (helpermr−1 prod)))

We prove the first equation mr−1 ◦mr = id .

Theorem 2. mr−1 ◦mr = id.

Proof. We show this by equational reasoning.

mr−1 ◦mr
= (eta-expansion)
λx → mr−1 (mr x )

= (def. mr)
λx → mr−1 (case x of (Producer o r)→ λ(ConsPar h)→ h o (ProdPar (mr r)))

= (case analysis on x )
λ(Producer o r)→ mr−1 (λ(ConsPar h)→ h o (ProdPar (mr r)))

= (def. mr−1)
λ(Producer o r)→ (λ(ConsPar h)→

h o (ProdPar (mr r))) (ConsPar (λx p → Producer x (helpermr−1 p)))
= (beta-reduction)

λ(Producer o r)→ Producer o (helpermr−1 (ProdPar (mr r)))

= (def. helpermr−1)

λ(Producer o r)→ Producer o (mr r (ConsPar (λx prod → Producer x (helpermr−1 prod))))
= (def. mr−1)
λ(Producer o r)→ Producer o (mr−1 (mr r))

= (coinduction hypothesis)
λ(Producer o r)→ Producer o r

= (def. id)
id

The second equation mr ◦mr−1 = id is more complicated since we can not do
case analysis on ProducerAlt , which is the function type ∀a.ConsPar o a → a.
We assume that values of this type can always be written using the constructor
function outputAlt. Where outputAlt is defined as:

outputAlt :: o → ProducerAlt o → ProducerAlt o
outputAlt o prod = λ(ConsPar cons)→ cons o (ProdPar prod)

Conjecture 1. Values of ProducerAlt can always be written in the form outputAlt o r .

This allows us to do a similar step to case analysis and complete the proof.

Theorem 3. mr ◦mr−1 = id.

Proof. We show this by equational reasoning.

mr ◦mr−1

= (eta-expansion)
λx → mr (mr−1 x )

= (def. mr−1)

λx → mr (x (ConsPar (λx p → Producer x (helpermr−1 p))))
= (Conjecture 1)

λ(outputAlt o r)→ mr ((outputAlt o r) (ConsPar (λx p → Producer x (helpermr−1 p))))
= (def. outputAlt)
λ(outputAlt o r)→ mr ((λ(ConsPar cons)→

cons o (ProdPar r)) (ConsPar (λx p → Producer x (helpermr−1 p))))
= (beta-reduction)

λ(outputAlt o r)→ mr ((λx p → Producer x (helpermr−1 p)) o (ProdPar r))
= (beta-reduction)

λ(outputAlt o r)→ mr (Producer o (helpermr−1 (ProdPar r)))

= (def. helpermr−1)

λ(outputAlt o r)→ mr (Producer o (r (ConsPar (λx prod → Producer x (helpermr−1 prod)))))
= (def. mr−1)
λ(outputAlt o r)→ mr (Producer o (mr−1 r))

= (def. mr)
λ(outputAlt o r)→ λ(ConsPar h)→ h o (ProdPar (mr (mr−1 r)))

= (coinduction hypothesis)
λ(outputAlt o r)→ λ(ConsPar h)→ h o (ProdPar r)

= (def. outputAlt)
λ(outputAlt o r)→ outputAlt o r

= (def. id)
id

B Additional Benchmarks

This section contains some additional benchmark results.

B.1 Deep-Pipe

The results of Spivey’s deep-pipe benchmark. This is a microbenchmark of the
merge operator, essentially a more extreme version of the primes benchmark.
The x-axis is logarithmic for readability of the graph.

Fig. 2. Results of the deep-pipe n benchmark.

B.2 Deep-Seq

The results of Spivey’s deep-seq benchmark. This is a microbenchmark of the
bind operator. The conduit library exhibits better performance compared to
pipes in this benchmark since it uses a Codensity encoding to improve the
performance of bind. The x-axis is logarithmic for readability of the graph.

Fig. 3. Results of the deep-seq n benchmark.

B.3 streaming-benchmarks

The results of some selected operation microbenchmarks from streaming-benchmarks

(https://github.com/composewell/streaming-benchmarks/). These bench-
marks measure the performance of specific operations such as map and fold , the
continuation-based implementation performs worse compared to the pipes and
conduit libraries. Despite this, the continuation-based approach is much more
performant in the use case of many merges as seen in the previous benchmarks.

Fig. 4. Results of various microbenchmarks.

https://github.com/composewell/streaming-benchmarks/

C ProducerAlt is representationally equal to ∀a.SP o a

We show the Producer case, an analogous strategy works for Consumer .

Theorem 4. The representation ProducerAlt is ∼R (representationally equiva-
lent) to ∀a.SP o a.

The relevant definitions for the proof are:

newtype ConsPar x a = ConsPar (x → ProdPar x a → a)
newtype ProdPar x a = ProdPar (ConsPar x a → a)
newtype SP o a = SP ((o → SP o a → a)→ a)
type ProducerAlt o = ∀a.ConsPar o a → a

Proof. First, we show that ConsPar o a → a ∼R SP o a.

ConsPar o a → a
∼R (def. ConsPar)

(o → ProdPar o a → a)→ a
∼R (def. ProdPar)

(o → (ConsPar o a → a)→ a)→ a
∼R (coinduction hypothesis)

(o → SP o a → a)→ a
∼R (def. SP)

SP o a

Then,

ProducerAlt o
∼R (def. ProducerAlt)
∀a.ConsPar o a → a

∼R (proven above)
∀a.SP o a ut

